
ABSTRACT 
RSS (Really Simple Syndication) is an XML-based standard that describes a simple framework for publishing 
headlines and links on the web. It’s widely used not only by major web sites, like CNN and the New York Times, but 
also by millions of individual bloggers. With an RSS Newsreader, you can easily and quickly gather headlines from 
your favorite web sites into one place, allowing you to scan a lot of different sources quickly. This paper shows how a 
SAS program can use the URL and XML access methods to collect RSS content from multiple web sites and present it 
in an easy-to-read format on a single web page. It may be of interest to anyone who uses SAS LE or Base SAS on the 
PC platform.  

INTRODUCTION 
Over the years, folks have grappled with the issue of how to best deal with the massive amount of content offered by 
the Web. For keeping tabs on information that changes continuously, such as news and news-like sites (e.g., personal 
weblogs), RSS (Really Simple Syndication) has emerged as one of the leading solutions.  
 
Widely used not only by major web sites, like CNN and the New York Times, but also by millions of individual 
bloggers, RSS is an XML-based standard that describes a simple framework for publishing headlines and links on the 
web. With an RSS Newsreader, you can easily and quickly gather headlines from all of your favorite web sites into one 
place, allowing you to scan a multitude of sources for items of interest. 
 
In the past year, the popularity of RSS has simply exploded, with companies like Yahoo, Google, AOL, and Microsoft 
all jumping on board with their own RSS offerings. And it looks like we may just be seeing the tip of the RSS iceberg. 
Other uses touted for RSS include political campaigns, project tracking, newsletters, calendaring, scheduling, multi-
media distribution, education, and even as a substitute for e-Mail (look Ma, no spam or viruses!). 

SAS AND RSS 
So how does SAS fit into the RSS picture?  Well, while the SAS URL Access Method made retrieving web-based data 
almost as easy as reading a sequential file, RSS introduced a whole new level of complexity with its XML tagsets and 
hierarchical data structures.  Fortunately, by introducing the XML Access Method, SAS was able to shield users from 
much of this ugly complexity by making it possible to decode the hierarchical XML data structures and map them into 
the relational data structures favored by SAS.  
 
A simple example of a SAS-based RSS Newsreader is presented in the following sections together with sample output 
to illustrate how the SAS URL and XML access methods can be used to retrieve multiple RSS news feeds from the 
web and consolidate the information into a single web page for viewing with Internet Explorer. 

SAS RSS NEWSREADER OVERVIEW 

1. Prepare a simple text file with a List of RSS Feeds that you wish to consolidate and view. 
URLs for RSS feeds can usually be found fairly easily at most major web sites, and are often 
identified by the cute orange-colored RSS logo. To make the RSS URLs easier to understand
at a later date, you can add blank lines and comments (* in column one) freely to the text file–
they will be ignored by the SAS Newsreader program.  

2. Provide an RSS XML Map to map individual RSS elements into SAS data sets & variables.    

3. The SAS Newsreader Program will: 
�� Use the URL access method to create a flat file on your PC for each of the web-based 

RSS Feeds you’ve specified 
�� Use the XML access method to create SAS data sets from each of the flat files  
�� Reformat and consolidate the information in the SAS data sets into a single HTML 

page 
�� Launch Internet Explorer 

4. Microsoft Internet Explorer is used to browse the consolidated news feeds, which may also 
include HTML formatting, links, and images. 

SAS RSS NEWSREADER CODE 
 

 

 

<?xml version="1.0" encoding="UTF-8"?> 
<SXLEMAP version="1.2" name="SXLEMap"> 
<!-- Create CHANNEL/ITEM SAS Datasets From RSS/RDF --> 
 
<TABLE name="CHANNEL"> 
<TABLE-PATH>//channel</TABLE-PATH> 
<TABLE-END-PATH beginend="Begin">//channel/item</TABLE-END-PATH> 
<COLUMN name="title"> 
 <PATH>//channel/title</PATH> 
 <TYPE>character</TYPE> 
 <DATATYPE>string</DATATYPE> 
 <LENGTH>32767</LENGTH> 
</COLUMN> 
<COLUMN name="link"> 
 <PATH>//channel/link</PATH> 
 <TYPE>character</TYPE> 
 <DATATYPE>string</DATATYPE> 
 <LENGTH>32767</LENGTH> 
</COLUMN> 
</TABLE> 
 
<TABLE name="ITEM"> 
<TABLE-PATH>//item</TABLE-PATH> 
<COLUMN name="title"> 
 <PATH>//item/title</PATH> 
 <TYPE>character</TYPE> 
 <DATATYPE>string</DATATYPE> 
 <LENGTH>32767</LENGTH> 
</COLUMN> 
<COLUMN name="URL"> 
 <PATH>//item/link</PATH> 
 <TYPE>character</TYPE> 
 <DATATYPE>string</DATATYPE> 
 <LENGTH>32767</LENGTH> 
</COLUMN> 
<COLUMN name="description"> 
 <PATH>//item/description</PATH> 
 <TYPE>character</TYPE> 
 <DATATYPE>string</DATATYPE> 
 <LENGTH>32767</LENGTH> 
</COLUMN> 
</TABLE> 
</SXLEMAP> 
 

 
* Consolidate & View RSS/RDF Feeds Using Internet Explorer; 
 
%let workdir=%sysfunc(pathname(work)); /* SAS Work Directory */   
%let tempdir=%sysfunc(pathname(temp)); /* Windows Temp Directory */   
 
*--> Create One Flat File From Each Feed Using URL Access Method; 
 
data _null_; 
length feed fvo $ 255.; 
infile "c:\RssFeeds.txt" truncover;      
input feed; 
if ^(feed=" "|feed=:"*");    
n+1; 
call symput('nfeeds',compress(put(n,5.))); 
fvo="&workdir\xmlworkxml"||compress(put(n,5.),' ')||".xml"; 
do while(^eof); 
  infile test url filevar=feed end=eof recfm=f; 
  input; 
  file out filevar=fvo noprint notitles recfm=f; 
  put _infile_; 
end; 
eof=0; 
run;  
  
*--> Create HTML From Each Flat File Using An XML Map; 
 
%macro xml2html; 
options noxwait; 
x erase "&tempdir\xmlwork.htm";             
%do i=1 %to &nfeeds; 
  libname test xml "&workdir\xmlworkxml&i..xml" xmlmap="c:\RssMap.map"; 
  data _null_; 
  file "&tempdir\xmlwork.htm" lrecl=32767 mod;       
  set test.channel; 
  put '<TABLE width=99% border=0 cellspacing=0 bgcolor=cccccc>' /  
      '<TR><TD><FONT size=3 color=000000>' / 
      '<a href="' link +(-1) '"><B>' title +(-1) '</B></a>' / 
      '</FONT></TD></TR></TABLE><FONT size=2>';  
  br='    '; 
  do while(^eof); 
   set test.item end=eof; 
   if title^="" then do;                      
     put br '<br><a href="' url +(-1) '"><b>' title +(-1) '</b></a>'; 
     br='<br>'; 
     end; 
   if description^="" then put br '<br>' description; 
   br='<br>'; 
  end; 
  put "<br><br></FONT>";  
  run;  
%end; 
%mend; 
%xml2html; 
 
*--> View Aggregated, Formatted News Feeds With Internet Explorer; 
 
options noxsync; 
x "&tempdir\xmlwork.htm"; 
options xsync; 

*** Techdirt *** 
http://www.techdirt.com/techdirt_rss.xml 
 
*** InternetNews.com *** 
http://headlines.internet.com/internetnews/top-news/news.rss 
 
*** Dan Gillmor - San Jose Mercury *** 
http://weblog.siliconvalley.com/column/dangillmor/index.xml 
 
*** ComputerWorld *** 
http://www.computerworld.com/news/xml/10/0,5009,,00.xml 

SAS RSS NEWSREADER OUTPUT 

CONCLUSION 
The URL and XML access methods offer SAS users the opportunity to explore the potential of RSS technology in a 
familiar (and productive!) setting.  
 
For the purposes of this paper, things have intentionally been kept relatively simple–use your imagination and you’re 
likely to find many more creative and useful ways to present and use RSS feeds. For example, the generated web 
page can be augmented with some JavaScript code to present the data in a more usable collapsible outline format. 
One might also want to explore the use of other RSS elements like pubdate (i.e., publication date), which could be 
used to limit viewing to those items that haven’t been seen before.  
 
Good luck! 

REFERENCES 
A wealth of RSS information, including tutorials and reference material, is freely available on the web.  
 
A good starting point is http://blogs.law.harvard.edu/tech/rss, which is edited by RSS pioneer Dave Winer, who co-
authored the RSS spec with Netscape. To keep abreast of the latest RSS developments, be sure to also check out 
Winer’s irreverent blog at http://www.scripting.com.  
 
For SAS-specific XML resources, including required XML LIBNAME engine enhancements for Release 8.2, visit the 
Base SAS Community at http://support.sas.com/rnd/base/index-xml-resources.html. 

